Une conique et ses tangentes

Reconnaître et caractériser l'ensemble défini dans le plan euclidien muni d'un repère $(O, \vec{\imath}, \vec{\jmath})$ par l'équation :

$$13x^2 - 32xy + 37y^2 - 5 = 0$$

Donner l'équation de la tangente au point $M_0(x_0, y_0)$. Chercher les points pour lesquels la tangente est parallèle à l'axe $(O, \vec{\imath})$.

Commençons par reconnaître la conique et effectuons en le tracé.

```
> load("reduction.mc")$
```

```
> (eq:13*x^2-32*x*y+37*y^2-5,Reduction(eq));
```

C'est une ellipse !

$$9\,Y^2 + X^2 = 1$$

```
> Elements();
```

$$\left[a = 1, b = \frac{1}{3}, e = \frac{2\sqrt{2}}{3}\right]$$

```
> RotationTranslation();
```

$$\left[\cos\varphi = \frac{2}{\sqrt{5}}, \sin\varphi = \frac{1}{\sqrt{5}}, X_\Omega = 0, Y_\Omega = 0\right]$$

```
> Representation(-2,-2,2,2,2);
```



$$37\,y^2 - 32\,x\,y + 13\,x^2 - 5 = 0$$

Partons à la recherche des équations des tangentes.

```
> (f(x,y):=13*x^2-32*x*y+37*y^2,'f(x,y)=f(x,y));
```

$$f\left(x,y\right)=37\,y^2-32\,x\,y+13\,x^2$$

```
> (g(x,y):=a*x+b*y,'g(x,y)=g(x,y));
```

$$g\left(x,y\right)=b\,y+a\,x$$

En résolvant le système $\begin{cases} f(x,y)=f(x_0,y_0) \\ g(x,y)=g(x_0,y_0) \end{cases}$ nous déterminerons les points d'intersection d'une droite et de l'ellipse passant toutes les deux par le point $M_0(x_0,y_0)$.

```
> sol:solve([f(x,y)-f(x0,y0),g(x,y)-g(x0,y0)],[x,y]);
```

$$\left[\left[x=x_0,y=y_0\right],\left[x=\frac{\left(32\,b^2+74\,a\,b\right)\,y_0+\left(37\,a^2-13\,b^2\right)\,x_0}{13\,b^2+32\,a\,b+37\,a^2},y=\frac{\left(13\,b^2-37\,a^2\right)\,y_0+\left(26\,a\,b+32\,a^2\right)\,x_0}{13\,b^2+32\,a\,b+37\,a^2}\right]\right]$$

La droite sera tangente à l'ellipse lorsque les deux points ci-dessus seront confondus.

```
> ab:solve([rhs(part(sol,2,1))-x0,rhs(part(sol,2,2))-y0],[a,b]);
```

$$\left[\left[a=R_1, b=-\frac{37\,R_1\,y_0-16\,R_1\,x_0}{16\,y_0-13\,x_0}\right], [a=0, b=0]\right]$$

La deuxième solution proposée, $(a=0, b=0)$, est à écarter — g ne caractériserait pas une droite —, il ne reste que la première dans laquelle figure une constante arbitraire R_1 que nous fixerons égale au dénominateur de l'expression de b.

```
> s:block([b],b:rhs(ab[1][2]),ab:subst(denom(b),%R1,ab[1]),radcan(ab));
```

$$\left[a=16\,y_0-13\,x_0, b=16\,x_0-37\,y_0\right]$$

Il ne reste qu'à écrire l'équation de la tangente.

```
> e:block(a:rhs(s[1]),b:rhs(s[2]),rat(g(x,y)-g(x0,y0),y0,x0,y,x)=0);
```

$$\left(-13\,x_0+16\,y_0\right)\,x+\left(16\,x_0-37\,y_0\right)\,y+13\,x_0^2-32\,y_0\,x_0+37\,y_0^2=0$$

On reconnaît le terme constant, il est égal à $f\,(x_0, y_0)$ lui même égal à 5. Voici donc le résultat recherché :

```
> a*x+b*y+5=0;
```

$$x\,\left(16\,y_0-13\,x_0\right)+y\,\left(16\,x_0-37\,y_0\right)+5=0$$

Nous aurions pu obtenir le même résultat avec la *règle du dédoublement*[1] en substituant $x x_0$ à x^2, $\frac{1}{2}(x y_0 + x_0 y)$ à $x y$ et $y y_0$ à y^2 dans l'équation de l'ellipse.

```
> (e:ev(eq,x^2=x*x0,y^2=y*y0),e:ratsubst(1/2*(x0*y+x*y0),x*y,e),e);
```

$$\left(37\,y-16\,x\right)\,y_0-16\,x_0\,y+13\,x\,x_0-5$$

Les substitutions sont faites, ordonnons.

```
> rat(e,y0,x0,y,x)=0;
```

$$\left(13\,x_0-16\,y_0\right)\,x+\left(-16\,x_0+37\,y_0\right)\,y-5=0$$

Nous retrouvons bien la même équation, au signe près.
Maintenant recherchons les points de l'ellipse où la tangente est parallèle à l'axe des abscisses c'est à dire lorsque $a = 0$.

```
> s:solve(a=0,x0);
```

$$\left[x_0=\frac{16\,y_0}{13}\right]$$

```
> y:(x0:rhs(s[1]), solve(f(x0,y0)-5));
```

[1] Qui se justifie bien avec le gradient.

$$\left[y_0 = -\frac{\sqrt{13}}{3\sqrt{5}}, y_0 = \frac{\sqrt{13}}{3\sqrt{5}}\right]$$

Nous avons les ordonnées des deux points attendus, voici les points.

```
> [subst(y[1],[x0,y0]),subst(y[2],[x0,y0])];
```

$$\left[\left[-\frac{16\sqrt{13}}{39\sqrt{5}}, -\frac{\sqrt{13}}{3\sqrt{5}}\right], \left[\frac{16\sqrt{13}}{39\sqrt{5}}, \frac{\sqrt{13}}{3\sqrt{5}}\right]\right]$$